```
>
```

```
> with(plots):
  with(DEtools):
> xphug:=R-> [diff(theta(t),t) = (v(t)^2 - cos(theta(t)))/(v(t)),
              diff(v(t),t)      = -sin(theta(t))-R*v(t)^2,
              diff(x(t),t)      = v(t)*cos(theta(t)),
              diff(y(t),t)      = v(t)*sin(theta(t))];
```

$$xphug := R \to \left[\frac{d}{dt}\,\theta(t) = \frac{v(t)^2 - \cos(\theta(t))}{v(t)},\ \frac{d}{dt}\,v(t) = -\sin(\theta(t)) - R\,v(t)^2,\ \frac{d}{dt}\,x(t) = v(t)\,\cos(\theta(t)),\ \frac{d}{dt}\,y(t) = v(t)\,\sin(\theta(t)) \right] \tag{1}$$

```
> Plot4 := proc(vmin, vmax, numv, {yrange:=0..4, R:=0, xrange:=-1..4})
   local inits,vi,i,cols;
   inits:= [seq( [theta(0)=0, v(0)=vi, x(0)=0, y(0)=vi],
     vi=vmin..vmax, (vmax-vmin)/numv )];
   cols:= [seq(COLOR(HUE,i),i=0..1,1/numv)];
   display(
   Array([DEplot(xphug(R), [theta,v,x,y], t=0..8,
          inits,
        theta=-Pi..3*Pi, v=0..3, x=xrange, y=yrange,
       linecolor=cols, numpoints=150, obsrange=false,
        scene=[theta,v]),
    DEplot(xphug(R), [theta,v,x,y], t=0..8,
          inits,
        theta=-Pi..3*Pi, v=0..3, x=xrange, y=yrange,
       linecolor=cols, stepsize=0.1, obsrange=false,
        scene=[x,y])]))
   end:
> Plot4(1,3,10,R=0.2,yrange=0..5,xrange=0..5);
```




```
> phug:=[ (v^2-cos(theta))/v, sin(theta)-R*v^2];
```

$$phug := \left[\frac{v^2 - \cos(\theta)}{v},\ \sin(\theta) - R\,v^2 \right] \tag{2}$$

```
> convert(solve({phug[1]=0, phug[2]=0}, {v,theta}), radical);
```

$$\left\{ \theta = \arctan\left(\sqrt{\frac{1}{R^2+1}}\ R,\ \sqrt{\frac{1}{R^2+1}} \right),\ v = \left(\frac{1}{R^2+1} \right)^{1/4} \right\} \tag{3}$$

```
> Fix:=R-> [arctan(sqrt(R^2/(1+R^2))), 1/(R^2+1)^(1/4) ];
```

$$Fix := R \to \left[\arctan\left(\sqrt{\frac{R^2}{1+R^2}} \right),\ \frac{1}{(1+R^2)^{1/4}} \right] \tag{4}$$

```
> Fix(0);Fix(1.2);
```

$$[0, 1]$$

$$[0.6550611021, 0.8001152415] \quad (5)$$

```
> xphug(R);
```

$$\left[\frac{d}{dt}\theta(t) = \frac{v(t)^2 - \cos(\theta(t))}{v(t)}, \frac{d}{dt}v(t) = -\sin(\theta(t)) - R\,v(t)^2, \frac{d}{dt}x(t) = v(t)\cos(\theta(t)), \frac{d}{dt}y(t) = v(t)\sin(\theta(t))\right] \quad (6)$$

From multivariable calculus, when I have

F (x,y) = < f(x,y), g(x,y) >

F(a+eps, b+delta) = F(a,b) + Matrix*<eps,delta> + O(eps^2, delta^2)
(The Matrix is the derivative matrix, or the Jacobian evaluated at (a,b)

If F(a,b) = (0,0), then this says that F(a+eps,b+delta) is approximately a matrix*(eps,delta).

So here, we need to look at the Jacobian.

```
> phugfunc:=unapply(phug,(theta,v));
```

$$phugfunc := (\theta, v) \to \left[\frac{v^2 - \cos(\theta)}{v}, \sin(\theta) - R\,v^2\right] \quad (7)$$

```
> J:=VectorCalculus[Jacobian](phugfunc(theta,v),[theta,v]);
```

$$J := \begin{bmatrix} \frac{\sin(\theta)}{v} & 2 - \frac{v^2 - \cos(\theta)}{v^2} \\ \cos(\theta) & -2\,R\,v \end{bmatrix} \quad (8)$$

can examine this matrix near fixed points.

```
> Fix(0);
```

$$[0, 1] \quad (9)$$

```
> eval(J,{theta=0,v=1,R=0});
```

$$\begin{bmatrix} 0 & 2 \\ 1 & 0 \end{bmatrix} \quad (10)$$

```
> Fix(2);evalf(%);
```

$$\left[\arctan\left(\frac{2}{5}\sqrt{5}\right), \frac{1}{5}5^{3/4}\right]$$

$$[0.7297276561, 0.6687403050] \quad (11)$$

```
> eval(J,{theta=0.7297,v=0.668,R=2});
```

$$\begin{bmatrix} 0.9979731328 & 2.670404168 \\ 0.7453744297 & -2.672 \end{bmatrix} \quad (12)$$

Claim it is worth understanding $\left(\frac{dx}{dt}, \frac{dy}{dt}\right) = A\,(x, y)$

for matrix A.

```
> A:=< < 2,0> | <0,-3> >;
```

$$A := \begin{bmatrix} 2 & 0 \\ 0 & -3 \end{bmatrix} \tag{13}$$

What is solution of DE (dx/dt, dy/dt) = A(x,y)?

Simpler: solve dx/dt = 2x

```
> dsolve(D(x)(t) = 2*x(t));
```

$$x(t) = \_C1\, e^{2t} \tag{14}$$

```
> < D(x)(t), D(y)(t)> = A . <x(t), y(t)>;
```

$$\begin{bmatrix} D(x)(t) \\ D(y)(t) \end{bmatrix} = \begin{bmatrix} 2\,x(t) \\ -3\,y(t) \end{bmatrix} \tag{15}$$

```
> DEplot( [ D(x)(t) = 2*x(t), D(y)(t) = -3*y(t)],
    [x(t),y(t)], t=-2..2,
     [seq( [x(0)=i, y(0)=1],i=-2..2, .2)],
     x=-2..2, y=-2..2, linecolor=black, obsrange=false);
```




```
> DEplot( [ D(x)(t) = 2*x(t), D(y)(t) = 3*y(t)],
    [x(t),y(t)], t=-2..2,
     [seq( [x(0)=i, y(0)=1],i=-2..2, .2)],
     x=-2..2, y=-2..2, linecolor=black, obsrange=false);
```




```
> MatToSys:=proc(A)
    local stuff;
    stuff:=A.<x(t),y(t)>;
    return( [ D(x)(t) = stuff[1], D(y)(t) = stuff[2]]);
  end:
> MatToSys( <<1,2>|<3,4>>);
```

$$[D(x)(t) = x(t) + 3\,y(t),\ D(y)(t) = 2\,x(t) + 4\,y(t)] \quad (16)$$

```
> DEplot( MatToSys(<<2,0>|<0,-3>>),
    [x(t),y(t)], t=-2..2,
     [seq( [x(0)=i, y(0)=1],i=-2..2, .2)],
     x=-2..2, y=-2..2, linecolor=black, obsrange=false);
```



```
> DEplot( MatToSys(<<2,1>|<0,-3>>),
    [x(t),y(t)], t=-2..2,
     [seq( [x(0)=i, y(0)=1],i=-2..2, .2),
      seq( [x(0)=i, y(0)=-1],i=-2..2, .2)],
     x=-2..2, y=-2..2, linecolor=black, obsrange=false);
```




```
> DEplot( MatToSys(<<2,1>|<1.6,-3>>),
    [x(t),y(t)], t=-2..2,
     [seq( [x(0)=i, y(0)=1],i=-2..2, .2),
      seq( [x(0)=i, y(0)=-1],i=-2..2, .2)],
    x=-2..2, y=-2..2, linecolor=black, obsrange=false);
```




```
> A:=<<2,1>|<16/10,-3>>
```

Warning, inserted missing semicolon at end of statement

$$A := \begin{bmatrix} 2 & \frac{8}{5} \\ 1 & -3 \end{bmatrix} \tag{17}$$

```
> with(LinearAlgebra):
> evalf(Eigenvectors(A));
```

$$\begin{bmatrix} 2.301785145 \\ -3.301785145 \end{bmatrix}, \begin{bmatrix} 5.301785149 & -0.3017851453 \\ 1. & 1. \end{bmatrix} \tag{18}$$

```
> JordanForm(A);
```

$$\begin{bmatrix} -\frac{1}{2} - \frac{1}{10}\sqrt{785} & 0 \\ 0 & -\frac{1}{2} + \frac{1}{10}\sqrt{785} \end{bmatrix} \tag{19}$$

```
> DEplot( MatToSys(<<0,2>|<-1,0>>),
    [x(t),y(t)], t=-2..2,
     [seq( [x(0)=i, y(0)=1],i=-2..2, .2),
```

```
seq( [x(0)=i, y(0)=-1],i=-2..2, .2)],
x=-2..2, y=-2..2, linecolor=black, obsrange=false);
```